use Kobayashi_Yokoyama_JJPS_replication.dta, clear

****************
* Preparations *
****************

/* Attach labels to variables and 
assign value labels to the variables. */

*Party identification
label variable pid "Party identification"
label define pid 0 "Independent" 1 "LDP supporter" 2 "DPJ supporter" 
label values pid pid

*Newspaper subscription
label variable newssub "Newspaper subscription"
label define newssub 0 "Neither subscriber" 1 "Yomiuri subscriber" 2 "Asahi subscriber" 3 "Both subscriber"
label values newssub newssub

*Experimental conditions: 1st vignette
label variable expcond_1stvign "Experimental conditions: 1st vignette"
label define expcond_1stvign ///
1 "LDP & DPJ" 2 "LDP" 3 "DPJ" 4 "No party cue" ///
5 "Yomiuri & Asahi" 6 "Yomiuri" 7 "Asahi" 8 "No press cue"
label values expcond_1stvign expcond_1stvign

*Experimental conditions: 2nd vignette
label variable expcond_2ndvign "Experimental conditions: 2nd vignette"
label define expcond_2ndvign ///
1 "Yomiuri & Asahi" 2 "Yomiuri" 3 "Asahi" 4 "No press cue" ///
5 "LDP & DPJ" 6 "LDP" 7 "DPJ" 8 "No party cue"
label values expcond_2ndvign expcond_2ndvign

****************
*** Table A1 ***
****************

/* Covariate balance among the entire sample of the pretreatment survey (N = 2,000) */

tab sex preexpcond, chi col
oneway age preexpcond, mean
oneway edu preexpcond, mean
tab fulltime preexpcond, chi col

****************
*** Table A2 ***
****************

/* Covariate balance among those who completed the posttreatment survey (N = 1,204) */

tab sex expcond_1stvign, chi col
oneway age expcond_1stvign, mean
oneway edu expcond_1stvign, mean
tab fulltime expcond_1stvign, chi col
oneway polint expcond_1stvign, mean
oneway cabapp expcond_1stvign, mean
tab supldp expcond_1stvign, chi col
tab supdpj expcond_1stvign, chi col
tab supnon expcond_1stvign, chi col
oneway ideology expcond_1stvign, mean

for var nhk - fnn: oneway X expcond_1stvign, mean
for var yomiuri - nikkei: oneway X expcond_1stvign, mean

oneway polknow expcond_1stvign, mean

****************
*** Table A3 ***
****************

/* Treatment effect of party cues on policy preferences */

*Temporary staffing services law
reg polpref_tssl ib4.expcond_1stvign##i.pid if expcond_1stvign < 5

*Right to collective self-defense
reg polpref_rcs ib8.expcond_2ndvign##i.pid if expcond_2ndvign > 4

****************
*** Table A4 ***
****************

/* Treatment effect of press cues on policy preferences */

*Temporary staffing services law
reg polpref_tssl ib8.expcond_1stvign##i.newssub if expcond_1stvign>4

*Right to collective self-defense
reg polpref_rcs ib4.expcond_2ndvign##i.newssub if expcond_2ndvign<5
